
APPENDIX B

COMPUTER CODES TO DETERMINE

THE AERODYNAMIC FORCES

During the HAWT operation, aerodynamic forces are significant part of the blade loads. Therefore, it is important that to determine this loads as possible as with high accuracy. In this study, the forces acting on each FEM mesh node is calculated by using the program FORCE which was coded in QBASIC programming language.

The program FORCE uses the pressure coefficient data which expressed with the Equation (2.17) to determine the aerodynamic forces. The pressure coefficient data is obtained from program AIRFOIL output which is coded by M. S. Kavsoglu. The program AIRFOIL uses “Smith and Hess Panel Method”. Panel is the small portion of the airfoil and seems like FEM element. The program produced pressure coefficients for each panel. So, pressure distribution around the airfoil can be obtained. In addition, the pressure distribution allows to calculate the force distribution around the airfoil. During the HAWT operation, different angles of attack are experienced by different blade sections. Therefore, an angle of attack spectrum should be considered for the force calculations. The program calculates pressure coefficients for a given angle of attack. So, pressure coefficient data is obtained for each angle of attack and put into the pressure coefficient matrix (rows for each panel, columns for each angle of attack). The program FORCE uses this matrix (as CP matrix). The program output includes the coordinates of panels of unit chord length airfoil. This data is used by both the program BLADE and the program FORCE.

If it is assumed that the X axis direction is parallel to the airfoil chord line, X and Y components of the force experienced by a panel are expressed with Equation (B.1).

$$F_X = 0.5 \rho \cdot C_{FX} \cdot (c \cdot h) \cdot W^2 \quad (\text{B.1a})$$

$$F_Y = 0.5 \rho \cdot C_{FY} \cdot (c \cdot h) \cdot W^2 \quad (\text{B.1b})$$

where ρ is the air density, c is the chord length of the blade section, h is the thickness of the blade segment, and W is the relative flow velocity.

Force coefficients C_{FX} and C_{FY} are obtained from the pressure coefficient of the panel as follows, and shown in Figure (B.1).

$$C_{FX} = D_Y \cdot K_{Pr} \quad (\text{B.2a})$$

$$C_{FY} = D_X \cdot K_{Pr} \quad (\text{B.2b})$$

D_X and D_Y are described as difference between X and Y coordinates of end points of the panel which is part of unit chord length airfoil respectively.

$$D_X = X_2 - X_1 \quad (\text{B.3a})$$

$$D_Y = Y_2 - Y_1 \quad (\text{B.3b})$$




Figure B.1 Pressure and force coefficients on an airfoil panel which has unit chord length.

The program FORCE uses five input data files. First and main data file name is entered after program run. Other file names are located into the main data file. Those

are dimensionless airfoil geometry data file obtained from program AIRFOIL output, dimensionless blade geometry data file obtained from program PALA output, pressure coefficient matrix data file obtained from program AIRFOIL output, and node coordinates of FEM mesh data file obtained from ANSYS.

Because program uses dimensionless airfoil and blade geometry data, radius of blade tip, center of twist and cone angle can be easily changed on the main data file according to the node coordinates data file. In addition, program allows to obtain blade loads for different wind speed and rotational speed by entering the data into the main data file only.

The program FORCE produced five output data files. The three of them include nodal force component values, F_x , F_y , and F_z if the cone angle is different from zero. Those files are used by an ANSYS macro to apply the aerodynamic loads on FEM model (see Appendix C). The fourth output file includes nodal pressure values which can be apply to the FEM model by using an ANSYS macro (see Appendix C). Fifth output file includes the sub-summation of force components for each blade section and total force components that acts on to whole blade.

The flow angle, θ , between relative flow and tangential velocity direction is calculated by two ways. Wake rotation is not considered in the first manner and program can calculate the relative flow angle according to the Figure (2.5) as follows,

$$\tan \theta = \frac{V}{r\Omega} \quad (\text{B.4})$$

According to the velocity diagram shown in Figure (2.5), relative flow can be expressed as follows,

$$W^2 = V^2 + (r\Omega)^2 \quad (\text{B.5})$$

In the second manner, wake rotation is considered and program can not calculate the angle (see Figure 2.8). In this case, the angle must be entered into blade geometry data file as fourth data of each section. So, program can calculate the relative flow velocity according to the velocity diagram shown in Figure (2.8) as follows,

$$W = V \sin\theta + r\Omega \cos\theta \quad (\text{B.6})$$

Important Program Parameter List

WS	Wind speed
W	Rotational speed
RV	Relative flow velocity
TWIST	Center of twist
CONE	Cone angle
FI	Relative flow angle
SETANG	Setting (or twisting) angle
ALPHA	Angle of attack
CTNMAX	Maximum number of FEM node
NDX,NDY,NDZ	Coordinates of a FEM node
AG()	Airfoil geometry data matrix
BG()	Blade geometry data matrix
CP()	Pressure coefficient data matrix
L1	Length of airfoil lower surface (for unit length)
L2	Length of airfoil upper surface (for unit length)
rhoinf	Air density
Pinf	Atmosphere pressure
P()	Nodal pressure
FX()	X components of nodal force
FY()	Y components of nodal force
FZ()	Z components of nodal force

```

*****PROGRAM FORCE*****
INPUT "INPUT DATA FILE NAME : ", IPT$
PRINT "PROGRAM FORCE.BAS IS RUNNING": PRINT
PRINT "Please wait..!": start! = TIMER
*****
'   DATA INPUT BLOCK
*****
      PI = 3.14159265359#
      OPEN IPT$ FOR INPUT AS #1
      LINE INPUT #1, RECS
      LINE INPUT #1, RECS
      INPUT #1, CTNMAX           'NODE NUMBER
      LINE INPUT #1, RECS
      INPUT #1, STARTROW        'ROW NUMBER OF NODE LIST FILE
                                'START BLOCK
      INPUT #1, BLOCK           'ROW NUMBER OF NODE LIST FILE
                                'DATA BLOCK
      INPUT #1, JUMP            'ROW NUMBER OF NODE LIST FILE
                                'BETWEEN DATA BLOCKS

      LINE INPUT #1, RECS
      INPUT #1, SN              'STATION NUMBER
      LINE INPUT #1, RECS
      INPUT #1, RAT              'RADIUS OF THE ROTOR
      LINE INPUT #1, RECS
      INPUT #1, TWIST            ' CENTER OF TWIST
      LINE INPUT #1, RECS
      INPUT #1, CONE             'CONE ANGLE [deg]
      CONE = CONE * PI / 180
      LINE INPUT #1, RECS
      INPUT #1, WS               'WIND SPEED [m/s]
      LINE INPUT #1, RECS
      INPUT #1, W                'ROTATIONAL SPEED [rad/s]
      LINE INPUT #1, RECS
      INPUT #1, RVTYPE           'TYPE OF RELATIVE VELOCITY
                                'CALCULATION: (1)SIMPLE
                                '(2)BY USING FLOW ANGLE FROM
                                'BLADE GEOMETRY FILE

      LINE INPUT #1, RECS
      INPUT #1, rhoinf           'AIR DENSITY [kg/m3]
      LINE INPUT #1, RECS
      INPUT #1, Pinf             'ATMOSPHERE PRESSURE [N/m2]
      LINE INPUT #1, RECS
      INPUT #1, PAR1             'SPANWISE MESH DIVISION OF SECTION
      LINE INPUT #1, RECS
      INPUT #1, PAR2             'HALF OF AIRFOIL MESH DIVISION
      LINE INPUT #1, RECS

```

```

INPUT #1, AAST          'ANGLE OF ATTACK VALUE OF
                        'FIRST CP MATRIX COLUMN

AAST = PI * AAST / 180
LINE INPUT #1, RECS$
INPUT #1, AAEND        'ANGLE OF ATTACK VALUE OF LAST CP
                        'MATRIX COLUMN

AAEND = PI * AAEND / 180
LINE INPUT #1, RECS$
INPUT #1, AAINC        'ANGLE OF ATTACK INCREMENT VALUE
                        'OF CP MATRIX

AAINC = PI * AAINC / 180
LINE INPUT #1, RECS$   '---INPUT FILES---
LINE INPUT #1, RECS$
INPUT #1, AIRFOIL$     'AIRFOIL GEOMETRY
LINE INPUT #1, RECS$
INPUT #1, BLADE$       'BLADE GEOMETRY
LINE INPUT #1, RECS$
INPUT #1, NLIST$       'NODE DATA
LINE INPUT #1, RECS$
INPUT #1, CP$          'CP MATRIX
LINE INPUT #1, RECS$   '---OUTPUT FILES---
LINE INPUT #1, RECS$
INPUT #1, PRSFILES$   'NODAL PRESSURE
LINE INPUT #1, RECS$
INPUT #1, FXFILES$    'FX COMPONENT OF NODAL FORCE
LINE INPUT #1, RECS$
INPUT #1, FYFILES$    'FY COMPONENT OF NODAL FORCE
LINE INPUT #1, RECS$
INPUT #1, FZFILES$    'FZ COMPONENT OF NODAL FORCE
LINE INPUT #1, RECS$
INPUT #1, FORCEFILES$  'TOTAL NODAL FORCE OF THE
                        'SECTIONS

CLOSE #1

```

```

****END OF INPUT BLOCK*****

```

```

'-----
'READ AIRFOIL GEOMETRY FROM A FILE
'-----
DIM AG(121, 2)      'AIRFOIL GEOMETRY (X,Y)
OPEN AIRFOIL$ FOR INPUT AS #1
LINE INPUT #1, RECS$
LINE INPUT #1, RECS$
LINE INPUT #1, RECS$
I = 1
DO WHILE NOT EOF(1)
    INPUT #1, X, Y
    AG(I, 1) = X

```

```

        AG(I, 2) = Y
        I = I + 1
LOOP
CLOSE #1
L1 = 0: L2 = 0

FOR J = 1 TO 120
DX = AG(J + 1, 1) - AG(J, 1)
DY = AG(J + 1, 2) - AG(J, 2)

IF J <= 60 THEN L1 = L1 + SQR(DX ^ 2 + DY ^ 2)
IF J > 60 THEN L2 = L2 + SQR(DX ^ 2 + DY ^ 2)

NEXT J
'-----
'READ BLADE GEOMETRY FROM A FILE
'-----
DIM BG(30, 4) 'BLADE GEOMETRY (C,BETA,R,FI)
OPEN BLADE$ FOR INPUT AS #1
LINE INPUT #1, RECS
I = 1
DO WHILE NOT EOF(1)
    INPUT #1, C, BETA, RR, FI
    BG(I, 1) = C * RAT
    BG(I, 2) = BETA
    BG(I, 3) = RR * RAT
    IF RVTTYPE = 1 THEN
        BG(I, 4) = ATN(WS / RR * RAT * W)
    ELSE IF RVTTYPE = 2 THEN
        BG(I, 4) = FI
    END IF
    I = I + 1
LOOP
CLOSE #1
RSTR = BG(1, 3)
REND = BG(SN, 3)
RINC = BG(2, 3) - BG(1, 3)
'-----
'READ CP MATRIX DATA FROM A FILE
'-----
'CPL= COLUMN NUMBER OF CP MATRIX
CPL = ABS(AAST - AAEND) / AAINC + 1
DIM CP(120, CPL)
OPEN CP$ FOR INPUT AS #1
LINE INPUT #1, RECS
LINE INPUT #1, RECS

```

```

LINE INPUT #1, RECS
LINE INPUT #1, RECS

I = 1
DO WHILE NOT EOF(1)
    LINE INPUT #1, RECS
    NCP = CINT(LEN(RECS) / 10)
    LNG = 1

    FOR J = 1 TO NCP
        CP(I, J) = VAL(MID$(RECS, LNG, 10))
        LNG = LNG + 10
    NEXT J
    I = I + 1
LOOP
CLOSE #1
'-----
'CALCULATE THE PRESSURES ON NODES
'-----
OPEN NLIST$ FOR INPUT AS #1
DIM FX(CTNMAX), FY(CTNMAX), FZ(CTNMAX), TX(90), TY(90)
DIM NPANEL(130, 90), CTR(CTNMAX), CT(90)
'DIM P(CTNMAX)
FOR I = 1 TO STARTROW
    LINE INPUT #1, RECS
NEXT I
PP = 1
DO WHILE NOT EOF(1)
    INPUT #1, NO, NDX, NDY1, NDZ1
    NDZ = NDZ1 * COS(CONE) + NDY1 * SIN(CONE)
    NDY = NDY1 * COS(CONE) - NDZ1 * SIN(CONE)
    NDZ = CINT(NDZ)
    FOR J = 2 TO SN
        IF NDZ < BG(J, 3) THEN
            STN = J - 1 'STATION NUMBER THAT NODE BELONGS TO
            GOTO OUTOF
        END IF
    NEXT J
OUTOF:
***CALCULATION OF CHORD LENGTH AT RADIUS OF THE NODE***
***BY USING LINEAR INTERPOLATION***
    STN1 = STN + 1
    BC = BG(STN, 1) - BG(STN1, 1)
    AD = BG(STN1, 3) - NDZ
    AB = BG(STN1, 3) - BG(STN, 3)
    CHLEN = BG(STN1, 1) + BC * (AD / AB) 'CHORD LENGTH
***CALCULATION OF SETTING ANGLE AT RADIUS OF THE NODE***

```

```

***BY USING LINEAR INTERPOLATION***
    BETA = BG(STN, 2) - BG(STN1, 2)
    SETANG = BG(STN1, 2) + BETA * (AD / AB) 'SETTING ANGLE
    SETANG = PI * SETANG / 180
***CALCULATION OF FLOW ANGLE AT RADIUS OF THE NODE***
    FI = BG(STN1, 4) + (BG(STN, 4) - BG(STN1, 4)) * (AD / AB)
    FI = PI * FI / 180
***CALCULATION ANGLE OF ATTACK AT RADIUS OF THE NODE***
    ALPHA = FI - SETANG
***COORDINATE SYSTEM TRANSFORMATION***
    X1 = NDX * COS(SETANG) + NDY * SIN(SETANG) + TWIST * CHLEN
    Y1 = NDY * COS(SETANG) - NDX * SIN(SETANG)
***COORDINATE TRANSFORMATION TO UNIT LENGTH***
    XO = X1 / CHLEN
    YO = Y1 / CHLEN
***SEARCH FOR PANEL NO THAT THE NODE BELONGS TO***
***PNO IS ROW NO OF THE CP MATRIX***
    IF YO <= 0 THEN
        FOR J = 2 TO 61
            IF XO > AG(J, 1) THEN
                PNO = J - 1      'PANEL NO THAT THE NODE BELONGS TO
                GOTO OUT1
            END IF
        NEXT J
    OUT1:
    ELSE
        FOR J = 62 TO 121
            IF XO < AG(J, 1) THEN
                PNO = J - 1      'PANEL NO THAT THE NODE BELONGS TO
                GOTO OUT2
            END IF
        NEXT J
    OUT2:
    END IF
***CALCULATION OF CP VALUE OF THE NODE BY USING LINEAR ***
*** INTERPOLATION CNO IS COLUMN NO OF THE CP MATRIX***
    A1 = ABS(ALPHA - AAST)
    DA = ((A1 / AAINC) - FIX(A1 / AAINC)) * AAINC
    A2 = A1 - DA
    CNO1 = (A2 / AAINC) + 1
    CNO2 = CNO1 + 1
    CB = CP(PNO, CNO1) - CP(PNO, CNO2)
***PRESSUE COEFFICIENT ON THE SPECIFIED NODE***
    CPNODE = CP(PNO, CNO1) - CB * (DA / AAINC)
***CALCULATION OF RELATIVE FLOW VELOCITY***
    IF RVTYPE = 1 THEN
        RV = SQR((WS^2) + (NDZ * W)^2)

```

```

ELSE IF RVTYPE =2 THEN
RV = NDZ * W * COS(FI) / 1000 + WS * SIN(FI)
END IF
***MAGNITUDE OF PRESSURE ON THE SPECIFIED NODE***
P(NO) = (CPNODE * (.5 * rhoinf * RV ^ 2) + Pinf) / 1000 'mN/mm^2 -
'Pascal/10e-3
'-----
' FORCE CALCULATION
'-----
'***QUADRILATERAL FEM ELEMENT SIZE
DS = (BG(STN1, 3) - BG(STN, 3)) / PAR1
IF PNO <= 60 THEN DL = L1 / PAR2
IF PNO > 60 THEN DL = L2 / PAR2
'-----
J = PNO
DO
J = J + 1
IF J > 120 THEN GOTO Sect1
X = AG(J, 1): Y = AG(J, 2)
L = SQR((XO - X) ^ 2 + (YO - Y) ^ 2)
LOOP WHILE L < DL / 2
'-----
Sect1:
I = PNO
DO
I = I - 1
IF I > 120 THEN GOTO Sect2
X = AG(I, 1): Y = AG(I, 2)
L = SQR((XO - X) ^ 2 + (YO - Y) ^ 2)
LOOP WHILE L < DL / 2
'-----
Sect2:
CTR(NO) = J - I - 1
JJ = 0
FOR II = RSTR TO REND STEP RINC / 3
JJ = JJ + 1
IF NDZ = CINT(II) THEN
GOTO Ext
END IF
NEXT II
Ext:
'-----
FL = 0: FD = 0
FOR K = I + 1 TO J - 1

FOR KK = 1 TO CT(JJ)
IF NPANEL(KK, JJ) = K THEN

```

```

CTR(NO) = CTR(NO) - 1
GOTO Double
END IF
NEXT KK
CTR = I
FOR P = CT(JJ) + 1 TO CT(JJ) + CTR(NO)
CTR = CTR + 1
NPANEL(P, JJ) = CTR
NEXT P

DX = AG(K + 1, 1) - AG(K, 1)
DY = AG(K + 1, 2) - AG(K, 2)
CP = CP(K, CNO1) - (CP(K, CNO1) - CP(K, CNO2)) * (DA / AAINC)
CFX = CP * DY
CFY = -CP * DX
FL = FL + .5 * rhoinf * CFX * (CHLEN * .001) * (DS * .001) * RV ^ 2
FD = FD + .5 * rhoinf * CFY * (CHLEN * .001) * (DS * .001) * RV ^ 2
Double:
NEXT K

FX(NO) = FL * COS(SETANG) - FD * SIN(SETANG)
FY(NO) = (FL * SIN(SETANG) + FD * COS(SETANG)) * COS(CONE)
FZ(NO) = -(FL * SIN(SETANG) + FD * COS(SETANG)) * SIN(CONE)
'-----
TFX = TFX + FX(NO)
TFY = TFY + FY(NO)
TFZ = TFZ + FZ(NO)

TX(JJ) = TX(JJ) + FX(NO)
TY(JJ) = TY(JJ) + FY(NO)

CT(JJ) = CT(JJ) + CTR(NO)
'-----
PP = PP + 1
IF PP = BLOCK + 1 THEN
    FOR TT = 1 TO JUMP
    IF NOT EOF(1) THEN
        LINE INPUT #1, RECS
    END IF
    NEXT TT
    PP = 1
END IF
LOOP
CLOSE #1

PRINT "TOTAL FX = ", TFX
PRINT "TOTAL FY = ", TFY

```

```

'-----
'WRITE PRESSURES AND FORCES ON FILE
'-----
'OPEN PRSFILE$ FOR OUTPUT AS #1
OPEN "FX.DAT" FOR OUTPUT AS #2
OPEN "FY.DAT" FOR OUTPUT AS #3
OPEN "FZ.DAT" FOR OUTPUT AS #4
FOR I = 1 TO CTNMAX
    'PRINT #1, P(I)          '[mN/mm2]
    PRINT #2, FX(I) * 1000 '[mN]
    PRINT #3, FY(I) * 1000 '[mN]
    PRINT #4, FZ(I) * 1000 '[mN]
NEXT I
'CLOSE #1
CLOSE #2
CLOSE #3
CLOSE #4
'-----
OPEN "FORCE.DAT" FOR OUTPUT AS #1

FOR I = 1 TO 88
PRINT #1, CT(I); " "; TX(I); " "; TY(I)
NEXT I

PRINT #1, " "
PRINT #1, "TOPLAM FX = ", TFX
PRINT #1, "TOPLAM FY = ", TFY
PRINT #1, "TOPLAM FZ = ", TFZ
CLOSE #1
'*****
finish! = TIMER
PRINT : PRINT "Program took"; finish! - start!; "seconds"

```

```

*****FORCE.BAS PROGRAM DATA INPUT FILE*****
**NODE NUMBER
5280
**ROW NUMBER OF NODE LIST FILE START BLOCK
5
**ROW NUMBER OF NODE LIST FILE DATA BLOCK
20
**ROW NUMBER OF NODE LIST FILE BETWEEN DATA BLOCKS
2
**STATION NUMBER
30
**RADIUS OF THE ROTOR [mm]
2400
**CENTER OF TWIST PERCENTAGE FROM LEADING EDGE
0.28
**CONE ANGLE [deg]
4
**WIND SPEED [m/s]
10.5
**ROTATIONAL SPEED [rad/s]
35
** TYPE OF REL. VEL. CALCULATION: (1)SIMPLE (2)READ FROM FILE
2
**AIR DENSITY [kg/m3]
1.225
**ATMOSPHERE PRESSURE [N/m2]
101325
**SPANWISE MESH DIVISION OF A SECTION
3
**HALF OF AIRFOIL BOUNDARY MESH DIVISION
30
**ANGLE OF ATTACK VALUE OF FIRST CP MATRIX COLUMN
2
**ANGLE OF ATTACK VALUE OF LAST CP MATRIX COLUMN
11
**ANGLE OF ATTACK INCREMENT VALUE OF CP MATRIX
0.5
-----INPUT FILES-----
**AIRFOIL GEOMETRY
23012.DAT
**BLADE GEOMETRY
PALA.DAT
**NODE DATA
NLIST.LIS
**CP MATRIX
CP23012.DAT

```

-----OUTPUT FILES-----
**NODE PRESSURE
BASINC.OPT
**FX COMPONENT OF NODE FORCE
FX.OPT
**FY COMPONENT OF NODE FORCE
FY.OPT
**FZ COMPONENT OF NODE FORCE
FZ.OPT
**TOTAL FORCE OF THE SECTIONS
FORCE.OPT